
Synopsis
An interactive, visual modeling tool designed to predict the growth of population and energy consumption—in the form of electricity, oil, and natural gas—based on data currently available from the 2008 CIA World Factbook.
Data Source

All calculations are based on pertinent data parsed from the 2008 edition of the CIA World Factbook. The fields used by this model are listed below, along with the corresponding field number. Each can be accessed at a URL of the following form by substituting the desired field number: 
https://www.cia.gov/library/publications/the-world-factbook/fields/[field number].html
#2119: Population

#2002: Population Growth Rate (%)
#2089: Industrial Production Growth Rate (%)
#2042: Electricity Consumption (kWh/year)
#2174: Oil Consumption (bbl/day)

#2181: Natural Gas Consumption (cu m/year)
Updating Data


The 2008 CIA World Factbook data has been parsed and compiled into a comma-separated data file (Factbook08.csv) readable by the modeling tool. This data file can be edited using Microsoft Excel, or by opening the CSV file in any text editor. The formatting of the CSV file must be strictly followed according to the field order indicated by the header row, and the header row must not be eliminated. The modeling tool accepts data input for up to 100 countries, and all rates must be provided as decimals and all energy values must be provided in units of quads using standard notation—meaning that scientific notation of the form 1.203E12 will produce an error. The following conversions have been provided for convenience: 
1 quad 
= 1 × 1015 BTU

= 2.93 × 1011 kWh of electricity

= 1.72 × 108 barrels of oil

= 2.77 × 1010 cubic meters of natural gas.
Calculations
Population for each country is calculated according to exponential growth using the 2008 value, the Population Growth rate, and the number of years elapsed. Electricity, oil, and natural gas consumption are calculated in a similar fashion using the Industrial Production Growth rate to predict rises in energy consumption. 
The width of each icon on the map is determined as the square root of the calculated population times an arbitrary scale factor. As a result, the area of each circle is proportional to the calculated population.
The color of each icon is determined based on the energy consumption of each country relative to the energy consumption of the greatest consumer at that time. The energy use of each country is divided by the energy use of the greatest consumer to arrive at an r-value, which is then used in a cube-root function to determine the display color: 

Source Code

The Visual Basic 6 project file (PopulationEnergyModel.vbp) has been provided along with all the supplementary files. The source code (reproduced below) has been well-documented and should be easily understandable to anyone with sufficient computer science background. In the event that questions arise, however, they may be directed to patel.sarav@gmail.com. The creator, Sarav Patel, provides said source code to Reza Toossi, for the purpose of enabling improvement and modification by future students under the explicit condition that the copyright of the original author not be removed. 
'Population and Energy Use Growth Model

'Copyright 2008 by Sarav Shailesh Patel

'Source code provided to Reza Toossi for the purpose of improvement and modification by future students under the explicit condition that the copyright of the original author not be removed.

Const Year0 = 2008                          'Starting year corresponding to all initial data

Const DataFile = "Factbook08.csv"           'Name of comma-separated data file

Private PerCapitaMode As Boolean            'TRUE: per capita energy calculations, FALSE: total energy calculations

Private Country(100) As String              'Country names

Private X(100), Y(100) As Integer           'Map coordinates for center of each country

Private PopulationRate(100), IndustrialRate(100) As Double 'Growth rates for population and energy use

Private Population0(100), Ele0(100), Oil0(100), Gas0(100) As Double 'Initial data for each country

Private Population(100), Ele(100), Oil(100), Gas(100), Ene(100), r(100) As Double 'Calculated data for each country

Private WorldPop, WorldEle, WorldOil, WorldGas, WorldEne As Double 'World totals

Private Sub Form_Load()

'Loads parameters for all countries from comma-separated data file

    Dim i As Integer

    Dim line, fields() As String

    Open DataFile For Input As #1

    For i = 0 To 100                        '0: header row of data file, up to 100 countries allowed

        Line Input #1, line

        fields() = Split(line, ",")         'Split into comma-separated parameters

        Country(i) = fields(0)              'Store information into the arrays

        X(i) = Val(fields(1))

        Y(i) = Val(fields(2))

        Population0(i) = Val(fields(3))

        PopulationRate(i) = Val(fields(4))

        IndustrialRate(i) = Val(fields(5))

        Ele0(i) = Val(fields(6))

        Oil0(i) = Val(fields(7))

        Gas0(i) = Val(fields(8))

        If i <> 0 Then Load shpIcon(i)      'Dynamically add an icon to the map at runtime for all countries (i = 0 is the header row)

        If EOF(1) Then Exit For

    Next i

    Close #1

    SetTotalMode

    hslYear.Value = Year0

End Sub

Private Sub hslYear_Change()

'Responds to a change in year by updating map

    lblYear.Caption = hslYear.Value

    CalcAll (hslYear.Value - Year0)         'Calculate all data based on the number of years passed since starting year

    lblWorldPop.Caption = FormatNumber(WorldPop, 0, , , True) 'Refresh world data

    lblWorldEle.Caption = Format(WorldEle, "0.000")

    lblWorldOil.Caption = Format(WorldOil, "0.000")

    lblWorldGas.Caption = Format(WorldGas, "0.000")

    lblWorldEne.Caption = Format(WorldEne, "0.000")

    For i = 1 To 100                        'Assign size and color to each icon based on calculated information

        If Country(i) = "" Then Exit For

        shpIcon(i).Width = Population(i) ^ (1 / 2) / 300 'Circle width is proportional to square root of population (area is proportional to population); 300 is scale factor

        shpIcon(i).Height = shpIcon(i).Width

        shpIcon(i).Left = imgMap.Left - shpIcon(i).Width / 2 + X(i) 'Center each icon at coordinates provided

        shpIcon(i).Top = imgMap.Top - shpIcon(i).Height / 2 + Y(i)

        shpIcon(i).BorderColor = RGB(0, r(i) * 255, r(i) * 255) 'Color assigned based on r-value (0: black, 1: cyan)

        shpIcon(i).Visible = True

    Next i

    imgMap.ZOrder 1                         'Execute "send to back" operation on map so that icons show on top

End Sub

Private Sub imgMap_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

'Determines the smallest icon currently highlighted at each mouse movement and refreshes map accordingly

    X = X / 15 + imgMap.Top                 'Twip to pixel conversion

    Y = Y / 15 + imgMap.Left

    Dim i, iSmallest, sizeSmallest As Integer

    iSmallest = 0                           'Stores index of the smallest icon currently highlighted

    sizeSmallest = 30000                    'Arbitrarily high starting value for the size of the smallest icon

    For i = 1 To 100

        If Country(i) = "" Then Exit For

        shpIcon(i).FillStyle = 1            'Set fill transparent for all icons

        If ((X - (shpIcon(i).Left + shpIcon(i).Width / 2)) ^ 2 + (Y - (shpIcon(i).Top + shpIcon(i).Height / 2)) ^ 2) < (shpIcon(i).Width / 2) ^ 2 Then 'Test if cursor is over icon

            If shpIcon(i).Width < sizeSmallest Then 'Test if icon is smaller than others under cursor

                sizeSmallest = shpIcon(i).Width

                iSmallest = i

            End If

        End If

    Next i

    If iSmallest <> 0 Then                  'Test if an icon is highlighted

        i = iSmallest

        shpIcon(i).FillColor = shpIcon(i).BorderColor 'Highlight icon

        shpIcon(i).FillStyle = 0

        lblCountry.Caption = Country(i)     'Refresh captions accordingly

        lblCountryPop.Caption = FormatNumber(Population(i), 0, , , True)

        lblCountryEle.Caption = Format(Ele(i), "0.000")

        lblCountryOil.Caption = Format(Oil(i), "0.000")

        lblCountryGas.Caption = Format(Gas(i), "0.000")

        lblCountryEne.Caption = Format(Ene(i), "0.000")

    Else

        lblCountry.Caption = ""             'Clear captions if no icon is highlighted

        lblCountryPop.Caption = ""

        lblCountryEle.Caption = ""

        lblCountryOil.Caption = ""

        lblCountryGas.Caption = ""

        lblCountryEne.Caption = ""

    End If

End Sub

Private Function CalcAll(ByVal T As Integer)

'Calculates all population and energy use data given the number of years since the reference year

    Dim i As Integer

    Dim maxEnergy As Double

    WorldPop = 0

    WorldEle = 0

    WorldOil = 0

    WorldGas = 0

    For i = 1 To 100

        If Country(i) = "" Then Exit For

        Population(i) = ExpGrowth(Population0(i), PopulationRate(i), T) 'Calculate population of each country

        Ele(i) = ExpGrowth(Ele0(i), IndustrialRate(i), T) 'Calculate energy use of each country

        Oil(i) = ExpGrowth(Oil0(i), IndustrialRate(i), T)

        Gas(i) = ExpGrowth(Gas0(i), IndustrialRate(i), T)

        Ene(i) = Ele(i) + Oil(i) + Gas(i)

        WorldPop = WorldPop + Population(i) 'Accumulate each country's data to the world totals

        WorldEle = WorldEle + Ele(i)

        WorldOil = WorldOil + Oil(i)

        WorldGas = WorldGas + Gas(i)

    Next i

    WorldEne = WorldEle + WorldOil + WorldGas

    If PerCapitaMode Then

        For i = 1 To 100

            If Country(i) = "" Then Exit For

            Ele(i) = Ele(i) / Population(i) * 10000000 'Calculate data per 10 million people for each country

            Oil(i) = Oil(i) / Population(i) * 10000000

            Gas(i) = Gas(i) / Population(i) * 10000000

            Ene(i) = Ene(i) / Population(i) * 10000000

        Next i

        WorldEle = WorldEle / WorldPop * 10000000 'Calculate data per 10 million people for world

        WorldOil = WorldOil / WorldPop * 10000000

        WorldGas = WorldGas / WorldPop * 10000000

        WorldEne = WorldEne / WorldPop * 10000000

    End If

    maxEnergy = 0

    For i = 1 To 100                        'Find country with greatest energy use

        If Country(i) = "" Then Exit For

        If Ene(i) > maxEnergy Then maxEnergy = Ene(i)

    Next i

    For i = 1 To 100

        If Country(i) = "" Then Exit For

        r(i) = (Ene(i) / maxEnergy) ^ (1 / 3) 'Assign each country an r-value based on energy use relative to the biggest user; cube root relationship gives better color transition than linear one

    Next i

End Function

Private Function ExpGrowth(ByVal Value0 As Double, ByVal Rate As Double, ByVal Time As Integer) As Double

'Calculates the exponential growth of a parameter based on initial value, rate, and years passed

    ExpGrowth = Value0 * (1 + Rate) ^ Time

End Function

Private Sub SetPerCapitaMode()

'Sets energy calculation mode to per capita and refreshes map

    PerCapitaMode = True

    lblEleUnits.Caption = "(quads per 10 million people)"

    lblOilUnits.Caption = "(quads per 10 million people)"

    lblGasUnits.Caption = "(quads per 10 million people)"

    lblEneUnits.Caption = "(quads per 10 million people)"

    hslYear_Change                          'Refresh map

End Sub

Private Sub SetTotalMode()

'Sets energy calculation mode to total and refreshes map

    PerCapitaMode = False

    lblEleUnits.Caption = "(quads)"

    lblOilUnits.Caption = "(quads)"

    lblGasUnits.Caption = "(quads)"

    lblEneUnits.Caption = "(quads)"

    hslYear_Change                          'Refresh map

End Sub

Private Sub lblCalcPerCapita_Click()

'Click event for per capita switch

    Shape2.Visible = True

    Shape1.Visible = False

    SetPerCapitaMode

End Sub

Private Sub lblCalcTotal_Click()

'Click event for total switch

    Shape1.Visible = True

    Shape2.Visible = False

    SetTotalMode

End Sub

